APPENDIX 5. MODELS OF THE EARTH'S MAGNETIC FIELD WITH INPUT PARAMETER FORMS

The following models of the earth's magnetic field are available. The input parameter forms, which describe the model, and the subroutine listings are given on the pages shown.

To add other models of the earth's magnetic field the user must write a subroutine that will calculate the normalized strength and direction of the earth's magnetic field (Y, Y_r, Y_θ, Y_φ) and their gradients ($\partial Y/\partial r$, $\partial Y/\partial\theta$, $\partial Y/\partial\varphi$, $\partial Y_r/\partial r$, $\partial Y_r/\partial\theta$, $\partial Y_r/\partial\varphi$, $\partial Y_\theta/\partial r$, $\partial Y_\theta/\partial\theta$, $\partial Y_\theta/\partial\varphi$, $\partial Y_\varphi/\partial r$, $\partial Y_\varphi/\partial\theta$, $\partial Y_\varphi/\partial\varphi$) as a function of position in spherical polar coordinates (r, θ, φ). ($Y = f_H / f$, where f_H is the electron gyrofrequency and f is the wave frequency.)

The restrictions on electron density models also apply to models of the earth's magnetic field. The coordinates r, θ, φ refer to the computational coordinate system, which is not necessarily the same as geographic coordinates. W24 and W25 give the geographic latitude and longitude of the north pole of the computational coordinate system.

The input to the subroutine (r, θ, ϕ) is through blank common. (See Table 3.) The output is through common block /YY/. (See Table 9.) It is useful if the name of the subroutine suggests the model to which it corresponds. It should have an entry point MAGY so that other subroutines in the program can call it. Any parameters needed by the subroutine should be input into W201 through W249 of the W array. (See Table 2.) If the subroutine needs massive amounts of data, these should be read in by the subroutine following the example of subroutine HARMONY.

INPUT PARAMETER FORM FOR SUBROUTINE CONSTY

An ionospheric model of the earth's magnetic field consisting of constant dip and gyrofrequency

Specify:

gyrofrequency, f_H = __________ MHz (W201)

dip, I = __________ degrees (W202)
radians

The magnetic meridian is defined by the geographic coordinates of the north magnetic pole:

latitude = __________ radians / degrees north (W24)

longitude = __________ radians / degrees east (W25)

```
      SUBROUTINE CONSTY                                                 CONY001
C         CONSTANT DIP AND GYROFREQUENCY                                CONY002
      COMMON /YY/ MODY,Y,PYPR,PYPTH,PYPPH,YR,PYRPR,PYRPT,PYRPP,YTH,PYTPRCONY003
     1,PYTPT,PYTPP,YPH,PYPPR,PYPPT,PYPPP                                CONY004
      COMMON /WW/ ID(10),W0,W(400)                                      CONY005
      EQUIVALENCE (F,W(6)),(FH,W(201)),(DIP,W(202))                     CONY006
      DATA (MODY=6HCONSTY)                                              CONY007
      ENTRY MAGY                                                        CONY008
      Y=FH/F                                                            CONY009
      YR=Y*SIN(DIP)                                                     CONY010
      YTH=Y*COS(DIP)                                                    CONY011
      RETURN                                                            CONY012
         END                                                            CONY013-
```

INPUT PARAMETER FORM FOR SUBROUTINE DIPOLY

An ionospheric model of the earth's magnetic field consisting of an earth centered dipole

The gyrofrequency is given by:

$$f_H = f_{H_0} \left(\frac{R_0 + h}{R_0}\right)^3 \left(1 + 3\cos^2\lambda\right)^{\frac{1}{2}}$$

The magnetic dip angle, I, is given by

$\tan I = 2 \cot \lambda$

h is the height above the ground

R_0 is the radius of the earth

λ is the geomagnetic colatitude

Specify:

the gyrofrequency at the equator on the ground, f_{H_0} = __________ MHz (W201)

the geographic coordinates of the north magnetic pole

latitude = __________ radians / degrees north (W24)

longitude = __________ radians / degrees east (W25)

```
      SUBROUTINE DIPOLY
      COMMON /CONST/ PI,PIT2,PID2,DUM(5)
      COMMON /YY/ MODY,Y,PYPR,PYPTH,PYPPH,YR,PYRPR,PYRPT,PYRPP,YTH,PYTPR
     1,PYTPT,PYTPP,YPH,PYPPR,PYPPT,PYPPP
      COMMON R(6) /WW/ ID(10),W0,W(400)
      EQUIVALENCE (EARTHR,W(2)),(F,W(6)),(FH,W(201))
      DATA (MODY=6HDIPOLY)
      ENTRY MAGY
      SINTH=SIN(R(2))
      COSTH=SIN(PID2-R(2))
      TERM9=SQRT(1.+3.*COSTH**2)
      T1=FH*(EARTHR/R(1))**3/F
      Y=T1*TERM9
      YR= 2.*T1*COSTH
      YTH= T1*SINTH
      PYRPR=-3.*YR/R(1)
      PYRPT=-2.*YTH
      PYTPR=-3.*YTH/R(1)
      PYTPT=.5*YR
      PYPR=-3.*Y/R(1)
      PYPTH=-3.*Y*SINTH*COSTH/TERM9**2
      RETURN
      END
```

INPUT PARAMETER FORM FOR SUBROUTINE CUBEY

A model of the earth's magnetic field consisting of a constant dip and a gyrofrequency which varies as the inverse cube of the distance from the center of the earth

This model has the same height variation as a dipole magnetic field.

The gyrofrequency is given by:

$$f_H = f_{Ho}\left(\frac{a}{r}\right)^3$$

a is the radius of the earth.

r is the distance from the center of the earth.

Specify:

gyrofrequency at the ground, f_{Ho} = ____________ MHz (W201)

dip, I = ____________ radians / degrees (W202)

The magnetic meridian is defined by the geographic coordinates of the north magnetic pole:

latitude = ____________ radians / degrees north / km (W24)

longitude = ____________ radians / degrees east / km (W25)

```
      SUBROUTINE CUBEY
C       CONSTANT DIP.
C       GYROFREQ DECREASES AS CUBE OF DISTANCE FROM CENTER OF EARTH.
C       THIS MODEL HAS SAME HEIGHT VARIATION AS A DIPOLE FIELD.
      COMMON /YY/ MODY,Y,PYPR,PYPTH,PYPPH,YR,PYRPR,PYRPT,PYRPP,YTH,PYTPR
     1,PYTPT,PYTPP,YPH,PYPPR,PYPPT,PYPPP
      COMMON /R /WW/ ID(10),W0,W(400)
      EQUIVALENCE (EARTHR,W(2)),(F,W(6)),(FH,W(201)),(DIP,W(202))
      DATA(MODY=5HCUBEY)
      ENTRY MAGY
       Y=(EARTHR/R)**3 *FH/F
      YR= Y*SIN(DIP)
      YTH= Y*COS(DIP)
      PYPR=-3.*Y/R
      PYRPR=-3.*YR/R
      PYTPR=-3.*YTH/R
      RETURN
         END
```

INPUT PARAMETER FORM FOR SUBROUTINE HARMONY

A model of the earth's magnetic field based on a spherical harmonic expansion

The upward, southerly, and easterly components of the earth's magneti field are given by:

$$H_r = -\sum_{n=o}^{6} (n+1)\left(\frac{a}{r}\right)^{n+2} \sum_{m=o}^{n} H_n^{\ m}(\theta)\left(g_n^{\ m} \cos m\varphi + h_n^{\ m} \sin m\varphi\right)$$

$$H_\theta = -\frac{1}{\sin\theta} \sum_{n=o}^{6} \left(\frac{a}{r}\right)^{n+2} \sum_{m=o}^{n} G_n^{\ m}(\theta)\left(g_n^{\ m} \cos m\varphi + h_n^{\ m} \sin m\varphi\right)$$

$$H_\varphi = \frac{1}{\sin\theta} \sum_{n=o}^{6} \left(\frac{a}{r}\right)^{n+2} \sum_{m=o}^{n} m\, H_n^{\ m}(\theta)\left(h_n^{\ m} \cos m\varphi - g_n^{\ m} \sin m\varphi\right)$$

where

a is the radius of the earth.

r, θ, φ are spherical (earth-centered) polar coordinates.

$H_o^{\ o}(\theta) = 1$

$H_1^{\ o}(\theta) = \cos\theta$

$H_1^{\ 1}(\theta) = \sin\theta$

$H_{m+1}^{\ m}(\theta) = H_m^{\ m}(\theta) \cos\theta$

$H_{m+1}^{\ m+1}(\theta) = H_m^{\ m}(\theta) \sin\theta$

$$H_{n+2}^{\ m}(\theta) = H_{n+1}^{\ m}(\theta) \cos\theta - \frac{(n+m+1)(n-m+1)}{(2n+3)\quad(2n+1)} H_n^{\ m}(\theta)$$

$$G_n^{\ m}(\theta) = -\frac{d}{d\theta} H_n^{\ m}(\theta) \sin\theta$$

$$G_m^{\ m}(\theta) = -m H_m^{\ m}(\theta) \cos\theta$$

$$G_{n+1}^{\ m}(\theta) = -(n+1) H_{n+1}^{\ m}(\theta)\cos\theta + \frac{(n+m+1)(n-m+1)}{2n+1} H_n^{\ m}(\theta)$$

The recursion formulas for calculating $H_n^{\ m}(\theta)$ and $G_n^{\ m}(\theta)$ are from Eckhouse (1964).

This subroutine uses coefficients $g_n^{\ m}$ and $h_n^{\ m}$ for Gauss normalization. Some coefficients are now being published for Schmidt normalization (e. g. Cain and Sweeney, 1970). The factors $S_{n,m}$ used for converting the "Schmidt normalized" coefficients to the "Gauss normalized" coefficients are as follows (Cain, et. al., 1968, Chapman and Bartels, 1940):

$$S_{o,o} = -1$$

$$S_{n,o} = S_{n-1,o} \left[\frac{2n-1}{n}\right]$$

$$S_{n,1} = S_{n,o} \sqrt{\frac{2n}{n+1}}$$

$$S_{n,m} = S_{n,m-1} \sqrt{\frac{(n-m+1)}{n+m}} \qquad \text{for } m>1$$

By convention, the "Gauss normalized" coefficient $g_1^{\ o}$ is positive, whereas the "Schmidt normalized" coefficient $g_1^{\ o}$ is negative. Coefficients based on more recent data on the earth's magnetic field including more satellite data are in the POGO 8/69 model.

Specify below the Gauss coefficients $g_n^{\ m}$ and $h_n^{\ m}$ in gauss.

	columns 2 → 10	columns 11 → 20	columns 21 → 30	columns 31 → 40	columns 41 → 50	columns 51 → 60	columns 61 → 70
1st card	$g_o^{\ o}$ = ______						
2nd card	$g_1^{\ o}$ = ______	g_1^1= _____					
3rd card	$g_2^{\ o}$ = ______	g_2^1= _____	g_2^2= _____				
4th card	$g_3^{\ o}$ = ______	g_3^1= _____	g_3^2= _____	g_3^3= _____			
5th card	$g_4^{\ o}$ = ______	g_4^1= _____	g_4^2= _____	g_4^3= _____	g_4^4= _____		
6th card	$g_5^{\ o}$ = ______	g_5^1= _____	g_5^2= _____	g_5^3= _____	g_5^4= _____	g_5^5= _____	
7th card	$g_6^{\ o}$ = ______	g_6^1= _____	g_6^2= _____	g_6^3= _____	g_6^4= _____	g_6^5= _____	g_6^6= _____

8th card	h_o^o = ______						
9th card	h_1^o = ______	h_1^1= ______					
10th card	h_2^o = ______	h_2^1= ______	h_2^2= ______				
11th card	h_3^o = ______	h_3^1= ______	h_3^2= ______	h_3^3= ______			
12th card	h_4^o = ______	h_4^1= ______	h_4^2= ______	h_4^3= ______	h_4^4= ______		
13th card	h_5^o = ______	h_5^1= ______	h_5^2= ______	h_5^3= ______	h_5^4= ______	h_5^5= ______	
14th card	h_6^o = ______	h_6^1= ______	h_6^2= ______	h_6^3= ______	h_6^4= ______	h_6^5= ______	h_6^6= ______

Set W200 = 1. to read in a set of coefficients.

This subroutine represents:

$H_n^m(\theta)$ by H(m+1, n+1)

$G_n^m(\theta)$ by G(m+1, n+1)

g_n^m by GG(m+1, n+1)

h_n^m by HH(m+1, n+1)

```
      SUBROUTINE HARMONY
C     MODEL OF THE EARTH S MAGNETIC FIELD BASED ON A HARMONIC ANALYSIS
      DIMENSION PHPTH(7,7),PGPTH(7,7),A1(7,7),B1(7,7)
      DIMENSION H(7,7),G(7,7),GG(7,7),HH(7,7),SINP(7),COSP(7)
      COMMON /YY/ MODY,Y,PYPR,PYPTH,PYPPH,YR,PYRPR,PYRPT,PYRPP,YTH,PYTPR
     1,PYTPT,PYTPP,YPH,PYPPR,PYPPT,PYPPP
      COMMON R(6) /WW/ ID(10),W0,W(400)
      COMMON /CONST/ PI,PIT2,PID2,DUM(5)
      EQUIVALENCE (THETA,R(2)),(PHI,R(3))
      EQUIVALENCE (EARTHR,W(2)),(F,W(6)),(READFH,W(200))
C     RATIO OF CHARGE TO MASS FOR ELECTRON
      DATA(EOM=1.7589E7)
      DATA (SET=0.),(H=1.,48(0.)),(G=49(0.)),(PHPTH=49(0.))
     1 ,(PGPTH=49(0.)),(MODY=7HHARMONY)
      ENTRY MAGY
      IF(SET) GO TO 2
      DO 1 M=1,7
      DO 1 N=1,7
      B1(M,N)=(N+M-1)*(N-M+1)/(2*N-1.)
    1 A1(M,N)=B1(M,N)/(2*N+1)
      SET=1.
```

```
    2 IF(READFH.EQ.0.) GO TO 3
      READ 2000,GG,HH
 2000 FORMAT(1X,F9.4,6F10.4)
      PRINT 2100,GG
 2100 FORMAT(1H1,10X,1H0,14X,1H1,14X,1H2,14X,1H3,14X,1H4,14X,1H5,14X,1H6
     1 /9X,7(1HG,14X)/10X,7(1HN,14X)//(1X,7F15.6))
      PRINT 2200,HH
 2200 FORMAT(//  11X,1H0,14X,1H1,14X,1H2,14X,1H3,14X,1H4,14X,1H5,14X,1H6
     1 /9X,7(1HH,14X)/10X,7(1HN,14X)//(1X,7F15.6))
      READFH=0.
 3    COSTHE=COS(THETA)
      SINTHE=SIN(THETA)
      AOR=EARTHR/R(1)
      PAORPR=-AOR/R(1)
      CNST2=AOR
      PCNSPR=PAORPR
      FIN1=PFIN1R=PFIN1T=PFIN1P=0.
      FIN2=PFIN2R=PFIN2T=PFIN2P=0.
      FIN3=PFIN3R=PFIN3T=PFIN3P=0.
      DO 4 M=1,7
      SINP(M)=SIN((M-1)*PHI)
 4    COSP(M)=COS((M-1)*PHI)
      H(1,2)=COSTHE
      H(2,2)=SINTHE
      DO 5 M=1,5
      H(M+1,M+2)=COSTHE*H(M+1,M+1)
      H(M+2,M+2)=SINTHE*H(M+1,M+1)
      DO 5 N=M,5
 5    H(M,N+2)=COSTHE*H(M,N+1)-A1(M,N)*H(M,N)
      DO 6 M=1,6
      G(M+1,M+1)=-M*COSTHE*H(M+1,M+1)
      PHPTH(M+1,M+1)=-G(M+1,M+1)/SINTHE
      PGPTH(M+1,M+1)=M*SINTHE*H(M+1,M+1)-M*COSTHE*PHPTH(M+1,M+1)
      DO 6 N=M,6
      G(M,N+1)=-N*COSTHE*H(M,N+1)+B1(M,N)*H(M,N)
      PHPTH(M,N+1)=-G(M,N+1)/SINTHE
 6    PGPTH(M,N+1)=N*SINTHE*H(M,N+1)-N*COSTHE*PHPTH(M,N+1)+B1(M,N)*PHPTH
     1 (M,N)
      DO 8 N=1,7
      CR=PCRPTH=PCRPPH=0.
      CTH=PCTHPT=PCTHPP=0.
      CPH=PCPHPT=PCPHPP=0.
      DO 7 M=1,N
      TEMP1=GG(M,N)*COSP(M)+HH(M,N)*SINP(M)
      TEMP2=(M-1)*(HH(M,N)*COSP(M)-GG(M,N)*SINP(M))
      CR    =CR    +H(M,N)*TEMP1
      PCRPTH=PCRPTH+PHPTH(M,N)*TEMP1
      PCRPPH=PCRPPH+H(M,N)*TEMP2
      CTH    =CTH    +G(M,N)*TEMP1
      PCTHPT=PCTHPT+PGPTH(M,N)*TEMP1
      PCTHPP=PCTHPP+G(M,N)*TEMP2
      CPH    =CPH    +H(M,N)*TEMP2
      PCPHPT=PCPHPT+PHPTH(M,N)*TEMP2
 7    PCPHPP=PCPHPP-H(M,N)*(M-1)**2*TEMP1
      CNST2=CNST2*AOR
      PCNSPR=CNST2*PAORPR+AOR*PCNSPR
      FIN1=FIN1+N*CNST2*CR
      PFIN1R=PFIN1R+N*PCNSPR*CR
      PFIN1T=PFIN1T+N*CNST2*PCRPTH
      PFIN1P=PFIN1P+N*CNST2*PCRPPH
      FIN2=FIN2+CNST2*CTH
      PFIN2R=PFIN2R+PCNSPR*CTH
      PFIN2T=PFIN2T+CNST2*PCTHPT
      PFIN2P=PFIN2P+CNST2*PCTHPP
      FIN3=FIN3+CNST2*CPH
```

```
      PFIN3R=PFIN3R+PCNSPR*CPH
      PFIN3T=PFIN3T+CNST2*PCPHPT
  8   PFIN3P=PFIN3P+CNST2*PCPHPP
      HTHETA=-FIN2/SINTHE
      HPHI=FIN3/SINTHE
C********* CONVERT FROM MAG FIELD IN GAUSS TO GYROFREQ IN MHZ
      CONST=-EOM/PIT2*1.E-6/F
      YR=-CONST*FIN1
      YTH=CONST*HTHETA
      YPH=CONST*HPHI
      Y=SQRT(YR**2+YTH**2+YPH**2)
      PYRPR=-CONST*PFIN1R
      PYTPR=-CONST*PFIN2R/SINTHE
      PYPPR=CONST*PFIN3R/SINTHE
      PYPR=(YR*PYRPR+YTH*PYTPR+YPH*PYPPR)/Y
      PYRPT=-CONST*PFIN1T
      PYTPT=-CONST*(PFIN2T/SINTHE+HTHETA*COSTHE/SINTHE)
      PYPPT=CONST*(PFIN3T/SINTHE-HPHI*COSTHE/SINTHE)
      PYPTH=(YR*PYRPT+YTH*PYTPT+YPH*PYPPT)/Y
      PYRPP=-CONST*PFIN1P
      PYTPP=-CONST*PFIN2P/SINTHE
      PYPPP=CONST*PFIN3P/SINTHE
      PYPPH=(YR*PYRPP+YTH*PYTPP+YPH*PYPPP)/Y
      RETURN
C   COEFFICIENTS IN GAUSSIAN UNITS FROM JONES AND MELOTTE (1953).
C   THE FOLLOWING 14 CARDS CAN BE USED AS DATA CARDS FOR THIS SUBROUTINE
C   0.
C.3039    .0218
C.0176   -.0509    -.0135
C-.0255   .0515    -.0236    -.0074
C-.0393  -.0397    -.0238     .0087    -.0018
C.0293   -.0329    -.0130     .0031     .0034     .0005
C-.0211  -.0073    -.0007     .0210     .0017    -.0004     .0006
C 0.
C        -.0555
C         .0260    -.0044
C         .0190    -.0033    -.0001
C        -.0139     .0076     .0019     .0010
C         .0057    -.0018     .0009     .0032    -.0004
C        -.0026    -.0204     .0018     .0009     .0004     .0002
C   THE FOLLOWING SET OF GAUSS NORMALIZED COEFFICIENTS WERE CONVERTED
C   FROM THE SCHMIDT NORMALIZED COEFFICIENTS CALCULATED BY LINEARLY
C   EXTRAPOLATING TO EPOCH 1974 THE COEFFICIENTS PUBLISHED FOR EPOCH
C  1960 BY CAIN AND SWEENEY (1970).  (USES EARTH RADIUS = 6371.2)
C .000000
C+.300953 +.020298
C+.028106 -.05214   -.014435
C-.0308   +.06560   -.025252  -.006952
C-.041243 -.043956  -.016897  +.008021  -.002525
C+.014742 -.037078  -.018906  +.002819  +.003656  +.000036
C-.006713 -.012234  -.004364  +.02137   +.001593  -.000072  +.00068
C .000000
C .000000 -.057886
C .000000 +.035942  +.001129
C .000000 +.011084  -.004421  +.001180
C .000000 -.010299  +.008794  -.000086  +.002256
C .000000 -.003849  -.012615  +.007845  +.002207  -.000328
C .000000 +.003157  -.012670  -.009281  +.002286  -.000135  +.000243
      END
```